@echo off
chcp 65001

set name=%1
set m=%2
set lib=%3

if  "%2" == "" (
    echo 用法: start build.bat 应用名 模块名 lib
    echo 参数《应用名》表示服务器上的应用文件夹名=微服务名
    echo 参数《模块》表示要部署的模块，对应的目录=模块/模块-api
    echo 参数《lib》 是固定字符串，表示 会删除服务器lib文件夹，重新上传
    pause
    exit 1;
)

set JAVA_HOME=D:\opt\Java\jdk-17

echo %name% - %m%
call mvn clean package -Dmaven.test.skip=true -e -U -am -P nacos -pl %m%

if %errorlevel% neq 0 exit /b %errorlevel%

echo ==============================
echo 打包 %name% - %m% 完成,拆包部署...
echo ==============================

cd %m%


@echo on
scp  target/*split-lib.jar  root@dc-app:/app/java/%name%/
@echo off

if  "%lib%" == "lib" (
    @echo on
    ssh root@dc-app 'rm -rf /app/java/%name%/lib ; mkdir -p /app/java/%name%/lib'
    scp -r target/split-lib/lib  root@dc-app:/app/java/%name%
    @echo off
)

cd ../..

echo %name% - %m% 上传完成！重启...
call restart-app.bat %name%

echo ==============================
echo %name% - %m% 上传重启完成！刷新浏览器 /health 接口，检查状态！
echo ==============================
